# ME201/MTH281/ME400/CHE400
# Heat Source in a Slab
# *Mathematica* 7

In this notebook we evaluate the solution of a problem of heat addition in a slab of finite thickness. A fixed amount of energy is added to the slab with an e-folding exponential time scale of $t^*$. We study how the maximum temperature in the slab varies as we vary $t^*$, with all other parameters remaining fixed. The formulation and scaling of the problem were done in class (section 5.2 of the notes) and will not be repeated here. The final result was that the scaled temperature is given by

```
temp[x_, t_] := β/(1 - β) * (Exp[-π^2 β t] - Exp[-π^2 t]) Sin[π x]
```

Here t is the time scaled by $L^2$/D, where $L$ is the width of the slab and $D$ is the diffusivity, and x is the length coordinate scaled by $L$. The temperature scaling was given in detail in class, but does not depend on the energy addition time. The parameter $\beta$ is the ratio of the diffusion time to the energy addition time, and is the main parameter of interest here. Our expectation is that as we decrease the energy addition time (i.e., make $\beta$ larger), the maximum temperature should increase, because the energy added has less time to diffuse away during the energy addition process.

We begin our analysis by noting that the location in space of the maximum temperature is always at $x$ = 1/2, where the sin function is 1. This maximum in space will depend on time, and we call it max[t]:

```
max[t_] = temp[1 / 2, t]
```

$$\frac{\left(-e^{-\pi^2 t} + e^{-\pi^2 t \beta}\right) \beta}{1 - \beta}$$

We observe that max[t] is zero for $t = 0$ and goes to zero as $t \to \infty$. Thus there will be a maximum in time at some finite non-zero $t$ which we call $t_M$. At the maximum, the derivative will be zero, and we use this to find $t_M$.

```
tm = First[Flatten[t /. Solve[D[max[t], t] == 0, t]]]
```

— Solve::ifun :
Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

$$-\frac{\text{Log}\left[\frac{1}{\beta}\right]}{\pi^2 \ (-1 + \beta)}$$

We ignore the warning. It is easy to obtain the same result by hand. Now we find the value of the maximum by evaluating max[t] at $t_M$. We call the maximum f[$\beta$].

```
f[β_] = max[tm]
```

$$\frac{\left(-\left(\frac{1}{\beta}\right)^{\frac{1}{-1+\beta}}+\left(\frac{1}{\beta}\right)^{\frac{\beta}{-1+\beta}}\right)\beta}{1-\beta}$$

Just a reminder that $\beta$ is the ratio of the diffusion time to the energy addition time. As we mentioned above, our expectation is that as we decrease the energy addition time (i.e., make $\beta$ larger), the maximum should increase, because the energy added has less time to diffuse away. We now check that expectation by examining the function f. First we look at large and small $\beta$ limits.

```
Limit[f[β], β → 0, Direction → -1]
```

0

The option Direction->-1 tells *Mathematica* to approach the point $\beta = 0$ from the right. We get zero for the limit, meaning that as the addition time gets arbitrarily large (compared with the diffusion time), there is no build-up of temperature at all. That is, if we add energy slowly enough, it leaks out so fast compared with the addition rate that there is no accumulation. Now we look at the opposite limit of large $\beta$.

```
Limit[f[β], β → ∞, Direction → 1]
```

1

This time we get 1. As we shall see shortly, this is the maximum value of the function f[$\beta$]. Thus when the energy addition time gets arbitrarily short compared with the diffusion time, we get the maximum build-up of temperature.

Although $\beta = 1$ appears to be a singularity, it is actually not. Both numerator and denominator vanish there, and the limit exists:

```
Limit[f[β], β → 1]
```

$$\frac{1}{e}$$

The limit is $e^{-1}$. Now finally we plot f[$\beta$].

```
Plot[f[β], {β, 0, 40}, AxesLabel → {"β", "Max T"},
 PlotLabel → "Max Temp Vs β", ImageSize → 360, PlotRange → {0, 1}]
```




This plot completely confirms our expectation.  The temperature build-up, as measured by the maximum temperature, increases monotonically with increasing ratio of diffusion time to energy addition time.

For any given set of parameter values, we could calculate the maximum temperature in the slab.  A more important point is that we have made the mathematics tell us something which agrees with our intuition, namely that what happens in the slab depends on the relative size of the two important time scales in the problem -- the energy addition time, and the diffusion time for heat loss.  When the energy addition time is much longer than the diffusion time, the heat leaks out of the slab as fast as we add it, and there is no temperature buildup (mathematically, $f[\beta]\to 0$ as $\beta\to 0$).  When the energy addition time is much shorter than the diffusion time, we get the largest possible temperature buildup for a given amount and spatial distribution of energy addition (mathematically, $f[\beta]$ has a maximum as $\beta\to\infty$).